
Summary 
Objectives: Meta-analysis allows to sum-
marize pooled sensitivities and specificities 
from several primary diagnostic test accuracy 
studies. Often these pooled estimates are in-
directly obtained from a hierarchical summary 
receiver operating characteristics (HSROC) 
analysis. This article presents a generalized 
linear random-effects model with the new 
SAS PROC GLIMMIX that obtains the pooled 
estimates for sensitivity and specificity di-
rectly. 

Methods Inf Med 2010; 49: 54–64 
doi: 10.3414/ME09-01-0001 
received: January 12, 2009 
accepted: July 29, 2009 
prepublished: November 20, 2009

Correspondence to: 
Jan Menke 
Gesellschaft für wissenschaftliche Datenverarbeitung 
Göttingen (GWDG) 
Am Faßberg 11 
37077 Göttingen 
Germany  
E-mail: jmenke@gwdg.de 

Methods: Firstly, the formula of the bivariate 
random-effects model is presented in context 
with the literature. Then its implementation 
with the new SAS PROC GLIMMIX is empiri-
cally evaluated in comparison to the indirect 
HSROC approach, utilizing the published 2 x 2 
count data of 50 meta-analyses. 
Results: According to the empirical evalu-
ation the meta-analytic results from the bi-
variate GLIMMIX approach are nearly iden -
tical to the results from the indirect HSROC 
approach. 
Conclusions: A generalized linear mixed 
model with PROC GLIMMIX offers a straight-
forward method for bivariate random-effects 
meta-analysis of sensitivity and specificity. 

1. Introduction 
Meta-analysis allows to summarize the 
 results from similar diagnostic test accu-
racy studies quantitatively. These primary 
studies compare an index test with its 
gold standard, resulting in dichotomous 
2   ×  2 contingency tables (�Table 1). Most 
meta-analyses report the diagnostic test 
accuracy by bivariate pairs of sensitivity 
and specificity, since these parameters can 
be readily interpreted. Sensitivity meas -
ures the probability of correctly diag -
nosing a disease or status, while speci -
 ficity measures the probability of cor  -
 rectly identifying its absence (�Table 1) 
[1–3].  

If several primary studies are conducted 
then their results usually differ. A meta-
analysis of these primary studies intends to 
find pooled summary estimates for sensi-
tivity and specificity that are more globally 
valid than the results of a single study. Such 
meta-analysis can be conducted with either 
a fixed-effects model or a random-effect 
model [4]. A fixed-effects model assumes 
that the variation among the primary 
studies originates from sampling errors 
alone [3, 4]. A random-effects model as-
sumes that such variation originates both 
from sampling errors and true differences 
between the primary studies, such as dif -
ferences in patient characteristics or other 
factors [3, 4].  

The bivariate random-effects model 
offers a straightforward meta-analytic ap-
proach for pooling sensitivity and specificity. 
Additionally it accounts for a possible cor -
relation between sensitivities and specifici -
ties of the included primary studies and 
for randomly distributed unspecified differ-
ences between the studies [2, 5]. Alterna -
tively, in the absence of covariates the hier-
archical summary receiver operating char-
acteristics (HSROC) model can be indi -
rectly used for bivariate meta-analysis by 
transforming the resulting HSROC esti-
mates to corresponding bivariate estimates 
[3, 6, 7].  

Several algorithms are used for bivariate 
meta-analysis [2, 3, 5, 6, 8  –14]. This article 
focuses on the implementation with the 
statistical software package SAS (SAS Insti-
tute, Cary, N.C.). The current SAS version 
9.2 incorporates PROC GLIMMIX that 
 implements a generalized linear mixed 
model. The purpose of this article is to 
present the application of PROC GLIM-
MIX for bivariate random-effects meta-
analysis of sensitivity and specificity. Addi-
tionally it is studied whether this GLIM-
MIX approach can be used instead of the 
indirect HSROC approach. 

Section 2 describes basics of mixed 
models; Section 3 specifies the bivariate 
mixed model for meta-analysis of sensitiv-
ity and specificity; Section 4 presents its 
 implementation with PROC GLIMMIX; 
Section 5 uses a meta-analytic standard 
example to compare the bivariate GLIM-
MIX approach to the indirect HSROC 
NLMIXED approach; Section 6 compares 
these approaches by using the 2   ×  2 contin-
gency data of 50 published meta-analyses; 
and Section 7 discusses the methods and 
results. An appendix with SAS code is 
supplemented. 

2. Mixed Models for  
Meta-analysis 

Currently SAS incorporates three different 
mixed models that can be used for fixed-
 effects and random-effects meta-analysis. 

2.1 Linear Mixed Model (LMM) 

PROC MIXED implements a linear mixed 
model (LMM). The observed data are 
 modeled directly as a linear combination of 
fixed effects, random effects, and sampling 
errors. In its general form the linear mixed 
model may be written as 
 
(Y – ε) = Xβ +                       Zu (2.1) 
 
or alternatively as   
 
E[Y | u] = Xβ +                       Zu 
 
                                       where Y is the vector of observed data, also 
termed the “response” of the model; ε is the 
vector of sampling errors; E [Y |   u] are the ex-
pected values of Y conditional on the ran-
dom effects u; X is the known design matrix 
for the fixed effects resulting from the 
MODEL statement in PROC MIXED; 
β is the vector of unknown fixed-effects 
 parameters; Z is the known design matrix for 
the random effects resulting from the RAN-
DOM statement in PROC MIXED; and u is 
the vector of unknown random  effects [15, 
16]. The fixed-effects and random-effects 
terms on the right-hand side of (2.1) are 
termed the linear predictor η (eta). The ran-
dom effects u and the sampling errors e are 
assumed to be uncorrelated and to be norm-
ally distributed with mean zero. Without the 
random-effects term Zu the linear mixed 
model reduces to a linear fixed-effects model. 
In contrast to a standard analysis of variance, 
PROC MIXED is not implemented with least 
squares methods. Instead, PROC MIXED 
utilizes iterative likelihood-based methods. 
PROC GLIMMIX and PROC NLMIXED are 
also implemented with likelihood-based 
methods and are described below. Usually 
(2.1) is written with its sampling errors ε on 
the right-hand side [15]. Here (2.1) is pres-
ented being a special case of the generalized 
linear mixed model. 

2.2 Generalized Linear Mixed 
Model (GLMM) 

PROC GLIMMIX implements a general-
ized linear mixed model (GLMM) that is a 
generalization of the linear mixed model 
(LMM). In addition to the linear mixed 
model’s properties, the GLMM has a link 
function that can account for different 
non-normal distributions of the observed 
Y data. The equation of the generalized 
 linear mixed model is given by 
 
                     g  (E[Y | u]) = Xβ + Zu (2.2) 
 
where on the left-hand side g(·) is a differ-
entiable monotonic link function with g   –1  (·) 
being the inverse link function; E[Y | u] are 
the expected values of Y conditional on the 
random effects u; and the terms on the right-
hand side are termed the linear predictor η 
(eta) with fixed-effects and random-effects 
terms as in (2.1) [16, 17]. Similar to PROC 
MIXED, PROC GLIMMIX also assumes a 
normal distribution for the unknown ran-
dom effects [16]. PROC GLIMMIX allows to 
use a logit link function for the observed 
2 × 2 contingency data of the diagnostic test 
accuracy studies, and to enter these data di-
rectly with an Events/Trial syntax. Without 
the random-effects term Zu the generalized 
linear mixed model reduces to a fixed-effects 
generalized linear model. If the link function 
g is the identity function, the generalized lin-
ear mixed model (2.2) reduces to the linear 
mixed model (2.1). 

2.3 Nonlinear Mixed Model 
(NLMM) 

PROC NLMIXED implements a nonlinear 
mixed model (NLMM) [15]. In this model 
the predictors are related to the observed 
data through a nonlinear function 
 
Y = f  (X, β                    , Z, u) + ε (2.3) 
 
where Y is the vector of observed data; f   (·) is 
a nonlinear function of fixed-effects com -
ponents X, β and random-effects com-
ponents Z, u; and ε is a vector of sampling 
 errors. Being an iterative method, PROC 
NLMIXED needs starting values that are not 
too different from the final model’s solution. 
The estimates from a nonlinear fixed-effects 
model may be used as such starting values 
for a subsequent nonlinear random-effects 
model. Due to its nonlinear capabilities 
PROC NLMIXED (NLMM) has more ap -
plications than the linear PROC MIXED 
(LMM) and PROC GLIMMIX (GLMM). 
The LMM and the GLMM are special linear 
cases of the NLMM and can thus also be 
 modeled with PROC NLMIXED. However, 
the numerical fitting of nonlinear models 
may be less stable than the fitting of linear 
 models [15]. Thus PROC MIXED and 
PROC GLIMMIX may have computational 
advantages over PROC NLMIXED, if they 
are appropriate for the studied statistical 
model. Additionally, PROC MIXED and 
PROC GLIMMIX offer more options for 
modeling the covariance structure of the 
random effects [16]. 

Table 1 2   ×  2 contingency table and diagnostic test accuracy parameters. TP (true-positives), FP 
(false-positives), FN (false-negatives), TN (true-negatives), RP (positive reference tests), RN (negative 
reference tests), IP (positive index tests), IN (negative index tests), N (total count), Sens (sensitivity), 
Spec (specificity), PPV (positive predictive value), NPV (negative predictive value); DOR (diagnostic odds 
ratio). If FP or FN is zero, DOR is not defined because of division by zero. A somewhat biased standard 
solution is to add 0.5 to all cells of the 2   ×  2 table, if any cell is zero [2, 3]. 

 Reference test   

positive negative total proportions 

Index positive TP FP IP = TP + FP PPV = TP / IP 

test negative FN TN IN = FN + TN NPV = TN / IN 

total RP = TP + FN RN = FP + TN N = RP + RN  

proportions Sens = TP / RP Spec = TN / RN DOR = (TP*TN) / (FP*FN) 

3. Specification of the 
 Bivariate Mixed Model 

The parameters of the bivariate mixed 
model are summarized in �Table 2. In a 
meta-analysis each primary study con-
tributes observed 2 × 2 count data, from 
which the observed proportions pA for 
 sensitivity and pB for specificity can be cal-
culated (�Table 1). These observed pro-
portions are usually affected with sampling 
errors and are thus only estimates for the 
true but unknown probabilities π. This can 
be written as 
 
E(pA, i) = πA, i  for sensitivity  
and (3.1) 
E(pB, i) = πB, i  for specificity of each study i             
 
with E(.) denoting the expected value of the 
random variable pA,i and pB,i, respectively. 

Sensitivity and specificity are binomial 
data. The relation between the 2 × 2 obser-
vations and the estimated probabilities π is 
a stochastical relation. This is denoted by a 
“~” sign:  
 
         sens: TPi ~ Binomial(RPi, πA, i),  
                                       with RPi = TPi + FNi 
 (3.2) 
spec: TNi ~ Binomial(RNi, πΒ, i),  
                                        with RNi = TNi + FPi 

 
Sensitivity and specificity can be linearized 
by transformation to the logit scale by 
(3.3): 
 
logit(π) = log(π  /(1– π)).                    (3.3) 
 
π = exp(logit(π))/(1 + exp(logit(π))) (3.4) 
 
�Equation 3.4 gives the inverse link. Fol-
lowing the notation of Harbord et al. [6], 

μA,i = logit(πA,i) is defined as the logit-
transformed estimated true sensitivity and 
μB,i = logit(πB,i) as the logit-transformed 
estimated true specificity in the ith study. 
μA and μB are the according summary sta -
tistics of sensitivity and specificity. These 
are bivariate fixed model effects that can 
represent all levels of diagnostic accuracy in 
a two-dimensional ROC space. These 
pooled summary statistics are usually esti-
mated and described by their mean and 
95%-CI (confidence interval). In a ran-
dom-effects meta-analysis the studies are 
allowed to contribute additional random 
effects of unknown origin to the model. 
 According to Harbord et al. the logit-trans-
forms μA,i and μB,i of the estimated true 
sensitivity and true specificity in each study 
are assumed to have a bivariate normal dis-
tribution across studies, thereby allowing 
for a possible correlation between them [6]. 
This bivariate random-effects model and 
its covariance structure has been specified 
by Reitsma et al. [5]: 
 

          

with   
(3.5)

 
 
In Equation 3.5 μAi and μBi are the logit-
transformed estimated true sensitivity and 
specificity of the ith study. Normal  (·) de-
notes the bivariate normal distribution. 
This bivariate normal distribution of 
 Equation 3.5 is determined by the bivari  -
 ate pooled mean estimates μA and μB for 
sensitivity and speci ficity, and by the bi-
variate covariance matrix ΣAB  . This co -
variance matrix includes the random-ef-
fects between-study variances σA

2 and σB
2 

of the studies’ sensitivities and specifici -
ties, and their covariance σAB. With the 
given definitions Equation 3.5 can be 
written as 
 
  
 
  

(3.6) 
This can also be written as 
 

, (3.7) 

Table 2  
Parameters for speci-
fication of the bivari-
ate mixed model. 

Element Description 

i denotes that the element belongs to study i 

TPi  , FPi  , FNi  , 
TNi 

observed true-positives, false-positives,  
false-negatives, and true-negatives of study i 

RPi = TPi + FNi observed positive reference tests of study i 

RNi = TNi + FPi observed negative reference tests of study i 

pA,i = TPi / RPi observed sensitivity (proportion) of study i 

pB,i = TNi / RNi observed specificity (proportion) of study i 

~ stochastical relation 

= equality relation 

Binomial  (n, π) binomial distribution with totals n and probability π 

Normal  (μ ,V) normal distribution with mean μ and variance structure V 

πA,i true sensitivity (probability) of study i 

πB,i true specificity (probability) of study i 

μA,i logit-transformed true sensitivity of study i 

μA 

μB 

UA,i 

UB,i 

σA
2 

σB
2 

σAB 

pooled sensitivity of the meta-analysis 

pooled specificity of the meta-analysis 

random effect for sensitivity of study i 

random effect for specificity of study i 

variance of the UA,i 

variance of the UB,i 

covariance of the UA,i and the UB,i 

μ B,i logit-transformed true specificity of study i

with 
  
 
 
 
In conjunction with Equation 3.2 this 
model (Eq. 3.7) is similar to the bivariate 
generalized linear mixed model as specified 
by Chu and Cole [11]. UA,i and UB,i are 
normally distributed random-effects pre-
dictions that may be correlated via their co-
variance σAB. Omitting UA,i and UB,i results 
in a fixed-effects model. 

4. Implementation of the 
Bivariate Mixed Model 
4.1 Data Structure 

The implementation of the bivariate model 
is shown for the historical lymphangi-
ography (LAG) data of Scheidler et al. [18]. 
In this example the data file LAG_DATA 
contains the 2 × 2 data of the 17 primary 
LAG studies with one record per study. 
LAG_BIDATA splits these data into two 
records per study and adds a status index to 

identify the bivariate data for sensitivity 
“A_pos” and specificity “B_neg” (�Fig. 1). 

4.2 Program Code of the Bivariate 
Mixed Model with PROC GLIMMIX  

The bivariate random-effects model of 
Equation 3.7 requires a generalized linear 
mixed model with the binomial link func-
tion (Eq. 2.2). This can be implemented with 
PROC GLIMMIX (�Fig. 2), where line 3 
specifies “study” and “status” as classification 
variables. In line 4 the 2   ×  2 count data of the 
primary studies are modeled directly with an 
Events/Trials syntax to speci fy the binomial 
proportions of sensitivity and specificity 
with the logit link as the default link. Line 5 
defines the random effects and generates the 
bivariate covariance matrix of Equation 3.7. 
“subject=study” iden tifies the studies as the 
groups, by which the covariance parameters 
are varied. Per study this gives a pair of ran-
dom-effects estimates for the status levels, 
i.e. one for sensitivity and one for specificity 
per study. “S” requests the output of these 
random-effects solutions. The covariance 
matrix of Equation 3.7 is modeled in terms 
of its Cholesky parameterization “type= 
chol”, as proposed by Jones and Huddleston 
[16]. If the iterative model estimation does 
not converge, “type=chol” can be replaced by 
“type=un”. “G” displays the estimated co-
variance matrix of the random effects. Lines 
6 and 7 give the pooled summary estimates 
μA and μB of sensitivity and specificity on the 
logit scale. Additionally, “ilink” requests their 
output on the original probability scale. 
Omitting line 5 gives the fixed-effects version 
of this bivariate generalized linear random-
effects model (�Appendix B and C). 

Fig. 1 Data code of the example in Section 4 

Fig. 2  
GLIMMIX code of the 
example in Section 4 

For personal or educational use only. No other uses without permission. All rights reserved.
Downloaded from www.methods-online.com on 2012-05-24 | IP: 38.107.179.232



Methods Inf Med 1/2010 © Schattauer 2010

58 J. Menke: Bivariate Random-effects Meta-analysis of Sensitivity and Specificity with SAS PROC GLIMMIX

5. Example 
As an example the described bivariate 
GLIMMIX approach of Section 4 was ap-
plied to the meta-analytic lymphangi-
ography data of Scheidler et al. [18]. For 
comparison the HSROC NLMIXED model 
of Macaskill was adapted to the terminol-
ogy of this article (�Appendix D and E) 
[3]. With the formulae of Harbord et al. this 
HSROC approach was used to indirectly 
calculate pooled estimates of sensitivity 
and specificity [6]. �Table 3 gives the re-
sults for the fixed-effects meta-analyses and 

�Table 4 gives the results of the random-
effects meta-analyses. In both tables the 
summary estimates and their confidence 
intervals are very similar, although they had 
been calculated with seemingly different 
models. �Figure 3 visualizes the results for 
the bivariate random-effects meta-analysis 
with PROC GLIMMIX in the two-dimen-
sional summary ROC space [19]. 

6. Empirical Model  
Comparison 
The bivariate GLIMMIX model was 
 empirically compared with the HSROC 
NLMIXED model by using the published 
2   ×  2 contingency data of 50 meta-analyses 
about diagnostic test accuracy [20–69]. 
These meta-analyses were identified from 
the literature by searching the PubMed 
data base for “meta-analysis”, “systematic 
review” and related terms from January 
2006 to March 2009. Per meta-analysis the 
2   ×  2 data of one index test were used, also if 
two or more index tests had been investi-
gated in that meta-analysis. From each 
meta-analysis those primary studies were 
excluded that had not investigated both 
sensitivity and specificity; for example if 
only diseased patients had been included so 
that the specificity of the index test could 
not be determined. The 2   ×  2 count data of 
the 50 meta-analyses were recalculated 
with the bivariate GLIMMIX model (�Ap-
pendix B and C) and with the indirect 
HSROC approach (�Appendix D and E). 
These meta-analytic models were com-
pared by regression analysis with the ques-
tion whether the GLIMMIX approach 
could be used instead of the indirect 

Table 3 Fixed-effects meta-analysis of the lymphangiography data. This table gives the mean (and 
95% confidence interval) of sensitivity and specificity for the fixed-effects meta-analysis with the 
 GLIMMIX and NLMIXED approaches (Appendix B, D).  

Table 4 Random-effects meta-analysis of the lymphangiography data. This table gives the mean 
(and 95% confidence interval) of sensitivity and specificity for the random-effects meta-analysis with 
the GLIMMIX and NLMIXED approaches (Appendix C, E).  

approach SAS PROC pooled sensitivity pooled specificity 

Bivariate GLIMMIX 67.1% (61.8–72.1) 80.0% (77.3–82.5) 

HSROC NLMIXED 67.1% (62.0–72.3) 80.0% (77.4–82.7)

approach SAS PROC pooled sensitivity pooled specificity 

Bivariate GLIMMIX 67.4% (59.8–74.2) 83.7% (75.1–89.8) 

HSROC NLMIXED 67.4% (60.2–74.6) 83.7% (76.5–91.0)

Fig. 3 Bivariate random-effects meta-analysis of sensitivity and specificity with PROC GLIMMIX. 
Summary ROC space of sensitivity and specificity. This figure visualizes the results of a bivariate random-
effects meta-analysis with PROC GLIMMIX for the lymphangiography (LAG) data of Scheidler et al. [18]. 
Rectangles: Observed bivariate pairs of sensitivity and specificity of the 17 primary LAG studies. Solid 
line: summary ROC curve. Central point: bivariate summary point. Ellipse: bivariate boundary of the 95% 
confidence region for the bivariate summary point. Dashed line: line of identity for sensitivity ver- 
sus (1 – specificity). The plot was generated with Review Manager (RevMan) version 5 [19]. Its input pa-
rameters were: E(logitSe) 0.7266; E(logitSp) 1.639; Var(logitSe) 0.125; Var(logitSp) 0.8233; Cov(logits) 
0.07664; SE(E(logitSe)) 0.1545; SE(E(logitSp)) 0.2505; Cov(Es) 0.004587; Studies 17. It makes no differ-
ence whether to use these bivariate model parameters or their HSROC equivalents: Lambda = Alpha 
2.1868; Theta 0.07407; beta 0.9491; Var(accuracy) 0.7797; Var(threshold) 0.1189. 

Fig. 4 Comparing fixed-effects models of the bivariate and the HSROC approach. Bivariate summary 
estimates of 50 meta-analyses: A) pooled mean sensitivities and specificities on the original probabil-
ity scale; B) their standard errors (SE) on the logit scale 

HSROC approach for pooling sensitivity 
and specificity.  

The bivariate GLIMMIX approach and 
the indirect HSROC NLMIXED approach 
converged in all of the 50 meta-analyses. 
The average calculation time was 1.4 sec-
onds per meta-analysis on a 2.6 GHz dual 
core Pentium computer in batch mode. 
�Figure 4A summarizes the pooled mean 
estimates for sensitivity and specificity of 
the fixed-effects versions of the meta-ana-
lytic models (r = 1, P = 0, standard devi-
ation (SD) of residuals = 0). �Figure 4B 
presents the corresponding standard errors 
of the pooled sensitivity and specificity 
(r = 1, P = 0, SD of residuals = 0). The 
 results of both fixed-effects models are 
iden tical. �Figure 5A shows that the 
pooled mean estimates of both random-ef-
fects models are nearly identical (r = 0.999, 
P <  0.001, SD of residuals = 0.016). �Fig -
ure 5B indicates that the corresponding 
standard errors are not all identical, but any 
differences are small (r = 0.999, P <  0.001, 
SD of residuals = 0.010). In summary, the 
bivariate GLIMMIX approach can be used 
instead of the indirect HSROC NLMIXED 
approach for calculating fixed-effects or 
random-effects summary estimates of sen-
sitivity and specificity. 

7. Discussion 

Diagnostic tests are valuable elements of the 
clinical diagnosis. In a screening population 
a diagnostic test intends to identify patients 
before they become symptomatic. In symp-
tomatic patients a diagnostic test supports 
the differential diagnosis and allows to assess 
the extend of a disease. However, similar to 
traditional diagnostic tests such as inspec-
tion and palpation, any other diagnostic test 
may also not be expected to be completely 
perfect. To judge the probability that a diag-
nostic test has given a true result it is necess-
ary to know the accuracy of that test [70]. 
The diagnostic test accuracy is often assessed 
by  several studies. A meta-analysis intends to 
quantitatively summarize such primary 
studies to obtain pooled test accuracy esti-
mates that are more globally valid than the 
results of one local study. 

Most meta-analyses of diagnostic test ac-
curacy studies report their results as bivariate 

pairs of sensitivity and specificity [20–69]. 
Sensitivity and specificity are based on bi-
nomial 2   ×  2 count data. The binomial dis-
tribution is a discrete distribution that ap-
proaches the Gaussian normal distribution 
only with large numbers. However, often the 
primary studies comprise small 2   ×  2 num -
bers. PROC GLIMMIX implements a gen-
eralized linear mixed model (GLMM, Section 
2.2) and considers the binomial distribution. 
Therefore, PROC GLIMMIX is in principle 
more appropriate for bivariate meta-analysis 
of sensitivity and specificity than PROC 
MIXED that implements a linear mixed 
model (LMM, Section 2.1) [8, 11, 71, 72]. 

Heterogeneity among the primary 
studies is a further issue to consider in a 

meta-analysis. The reported sensitivities 
and specificities of the primary studies 
usually differ, at least because of sampling 
errors. However, often there is more disper-
sion (overdispersion) among the studies’ 
sensitivities and/or specificities than what 
may be expected from sampling errors 
alone [12, 13, 73]. Such excessive dispersion 
is termed “heterogeneity” and can be as-
sessed by the Q-test and the I2-statistics that 
are provided by the free meta-analysis pro-
gram Meta-DiSc [74, 75].  

Heterogeneity may arise from unknown 
differences in the study populations or 
other factors. A random-effects model ac-
counts for such differences among the pri-
mary studies. By its covariance matrix it 

Fig. 5 Comparing random-effects models of the bivariate and the HSROC approach. Bivariate sum-
mary estimates of 50 meta-analyses: A) pooled mean sensitivities and specificities on the original prob-
ability scale; B) their standard errors (SE) on the logit scale 

also accounts for a possible correlation be-
tween sensitivity and specificity on the 
study level [6]. A fixed-effects model is 
closely related to the random-effects 
model. It can be regarded as a special case of 
a random-effects model with the random 
effects being zero [3]. However, many 
meta-analyses observe some heterogeneity 
among their primary studies and thus 
apply a random-effects model [12, 13]. 

The HSROC approach (�Appendix E) 
accounts for random effects and be- 
longs to the more rigorous meta-analytic 
methods [3, 6, 10, 76]. Macaskill has shown 
that the empirical Bayes estimates for sensi-
tivity and specificity from the HSROC 
NLMIXED approach agree closely with 
those of a full Bayesian analysis [3, 10]. The 
HSROC approach models the 2   ×  2 count 
data of the primary studies in terms of 
diagnostic accuracy (alpha) and threshold 
(theta). The additional scale parameter 
(beta) provides for asymmetry in the sum-
mary ROC curve by allowing accuracy to 
vary with threshold [3]. However, most 
meta-analyses prefer to report their results 
in terms of sensitivity and speci ficity 
[20–69]. With the formulae given by Har-
bord et al. the HSROC approach can be 
used to calculate the pooled sensitivity and 
specificity indirectly [6]. 

This study has shown that the bivariate 
GLIMMIX approach offers a straightfor-
ward alternative to the indirect HSROC ap-
proach. The empirical evaluation of 50 
meta-analyses has shown that the results of 
both meta- analytic approaches are very 
similar, although being calculated with 
seemingly different models. Thus the pre -
sented GLIMMIX approach has the same 
accuracy as the indirect HSROC approach. 
The direct modeling of sensitivity and 
 specificity may be considered an advantage 
of the bivariate GLIMMIX approach. A 
further advantage of the GLIMMIX ap-
proach is its brief program code that ap-
pears to be more comprehensible and more 
structured than the code of the indirect 
HSROC NLMIXED approach (�Appen-
dix B to E). 

This study is limited to comparing the 
bivariate GLIMMIX approach and the in-
direct HSROC NLMIXED approach, 
which are mathematically equivalent ap-
proaches for the bivariate meta-analysis of 

sensitivity and specificity [6]. In the em-
pirical evaluation of the 50 meta-analyses 
(Section 6) the computational perform-
ance of these both SAS implementations 
was similar with fast numerical conver-
gence in all cases [20–69]. A comparison 
with the different other published meta-
analytic approaches was beyond the scope 
of this study [2, 3, 5, 8–14]. 

This article has introduced the general-
ized linear mixed model with PROC 
GLIMMIX for bivariate random-effects 
meta-analysis of sensitivity and specificity. 
The model can also be used for fixed-effects 
meta-analysis as shown in Sections 4 to 6. 
Furthermore, the  bivariate GLIMMIX ap-
proach can be used to obtain the HSROC 
results when utilizing the equations of Har-
bord et al. [6]. The results from the bivari-
ate GLIMMIX approach can be applied 
with the RevMan program to plot bivariate 
summary estimates and SROC curves, as 
shown in �Figure 3 [19]. Among further 
applications are the bivariate meta-regres-
sion and the bivariate comparison of differ-
ent index tests. For these purposes the 
PROC MIXED codes of van Houwelingen 
et al. and Reitsma et al. can be well adapted 
to the PROC GLIMMIX syntax [2, 5]. 

In summary, a generalized linear ran-
dom-effects model with PROC GLIMMIX 
offers a straightforward method for bivari-
ate meta-analysis of sensitivity and specifi-
city, and is thus an alternative to the closely 
related indirect HSROC approach. 

Appendix 
ods html body='output.htm' style=statistical; 
/***************************************************************/ 
/* A. Data Files                                                            */ 
/***************************************************************/ 
%let nStudies = 17; 
%let n_minus1 = 16;  
/*     n_minus1 is used for degrees of freedom in fixed-effects  
  models to accomplish for the fact that LAG_bidata contains 34 
  records although these represent only 17 different studies.  
  In the random-effects models this is automatically considered 
  by the 'random ... subject=study' statements. */ 
DATA LAG_data;  
length author $20;  
input number author $ year TP FP FN TN;  
RP=TP+FN; RN=TN+FP; 
study = _N_; 
datalines; 
11 Kindermann 1970 19  1 10  81 
12 Lecart 1971  8  9  2  13 
13 Piver 1971 41  1 12  49 
14 Piver 1973  5  1  2  18 
15 Kolbenstvedt 1975 45 58 32 165 
16 LemanJr 1975  8  6  2  32 
17 Brown 1979  5  8  1   7 
18 Lagasse 1979 15 17 11  52 
19 Kjorstad 1980 16 11  8  24 
10 Ashraf 1982  4  8  2  25 
11 deMuylder 1984  8 12 10  70 
12 Smales 1986 10  4  4  55 
13 Feigen 1987  2  5  6  23 
14 Swart 1989  7 10  7  30 
15 Heller 1990 44 50 12 135 
16 Lafianza 1990  8  3  1  37 
17 Stellato 1992  4  3  0  14 
;run; PROC PRINT data=LAG_data;run; 
 
DATA LAG_bidata; set LAG_data; 
status='A_pos'; true=TP; pos=TP; total=RP; output; 
status='B_neg'; true=TN; pos=FP; total=RN; output; 
keep study status true pos total; run; 

/***************************************************************/ 
/* B. Bivariate generalized linear fixed-effects model */ 
/***************************************************************/ 
PROC GLIMMIX data=LAG_bidata method=quad; 
title 'Bivariate generalized linear fixed-effects model'; 
class study status; 
model true/total=status/noint s cl corrb covb df=&n_minus1; 
estimate 'logit_sens' status 1 0 / cl ilink;  
estimate 'logit_spec' status 0 1 / cl ilink;  
estimate 'LOR' status 1 1 / cl exp;  
run; 
 
/***************************************************************/ 
/* C. Bivariate generalized linear random-effects model */ 
/***************************************************************/ 
PROC GLIMMIX data=LAG_bidata method=quad; 
title 'Bivariate generalized linear random-effects model'; 
class study status; 
model true/total = status / noint s cl corrb covb ddfm=bw; 
random status / subject=study S type=chol G; 
/* if the model does not converge then replace 'chol' by 'un' */  
estimate 'logit_sens' status 1 0 / cl ilink;  
estimate 'logit_spec' status 0 1 / cl ilink;  
estimate 'LOR' status 1 1 / cl exp;  
run; 
 
/***************************************************************/ 
/* D. SROC nonlinear fixed-effects model */ 
/***************************************************************/ 
PROC NLMIXED data=LAG_bidata tech=quanew df=&n_minus1; 
title 'SROC nonlinear fixed-effects model'; 
parms Theta=0 Alpha=0; 
if (status='A_pos') then eta = Theta + Alpha/2; 
if (status='B_neg') then eta = Theta - Alpha/2; 
pi = exp(eta)/(1+exp(eta)); 
model pos ~ binomial(total,pi); 
/* calculate bivariate parameters */ 
mu_A = Alpha/2 + Theta; 
mu_B = Alpha/2 - Theta; 
/* Bivariate model parameters */ 
estimate 'logit_sens' mu_A; 
estimate 'logit_spec' mu_B; 
estimate 'LOR' mu_A + mu_B; 
estimate 'sens' exp(mu_A)/(1+exp(mu_A)); 
estimate 'spec' exp(mu_B)/(1+exp(mu_B)); 
estimate 'DOR' exp(mu_A + mu_B); 
run; 

/***************************************************************/ 
/* E. HSROC nonlinear random-effects model */ 
/***************************************************************/ 
PROC NLMIXED data=LAG_bidata tech=quanew; 
title 'HSROC nonlinear random-effects model'; 
parms Theta=0 Alpha=0 Beta=0 s2ut=0 s2ua=0; 
bounds s2ut >= 0, s2ua >= 0; 
if (status='A_pos') then eta = exp(-Beta/2)*(Theta+ut +(Alpha+ua)/2); 
if (status='B_neg') then eta = exp( Beta/2)*(Theta+ut -(Alpha+ua)/2); 
pi = exp(eta)/(1+exp(eta)); 
model pos ~ binomial(total,pi); 
random ut ua ~ normal([0,0],[s2ut,0,s2ua]) subject=study; 
/* calculate bivariate parameters */ 
mu_A = exp(-Beta/2) * (Alpha/2 + Theta); 
mu_B = exp( Beta/2) * (Alpha/2 - Theta); 
var_uA = exp(-Beta) * (s2ut + s2ua/4); 
var_uB = exp( Beta) * (s2ut + s2ua/4); 
cov_uAB = s2ua - s2ut/4; 
/* Bivariate model parameters */ 
estimate 'var_uA' var_uA; 
estimate 'var_uB' var_uB; 
estimate 'cov_uAB' cov_uAB; 
estimate 'logit_sens' mu_A; 
estimate 'logit_spec' mu_B; 
estimate 'LOR' mu_A + mu_B; 
estimate 'sens' exp(mu_A)/(1+exp(mu_A)); 
estimate 'spec' exp(mu_B)/(1+exp(mu_B)); 
estimate 'DOR' exp(mu_A + mu_B);  
run; 
 
ods html close;

